<commit_before>{ stdenv, fetchgit, talloc, enableStatic ? false }:

stdenv.mkDerivation rec {
  name = "proot-${version}";
  version = "4.0.3";

  src = fetchgit {
    url = "git://github.com/cedric-vincent/proot.git";
    rev = "refs/tags/v${version}";
    sha256 = "95a52b2fa47b2891eb2c6b6b0e14d42f6d48f6fd5181e359b007831f1a046e84";
  };

  buildInputs = [ talloc ];

  preBuild = stdenv.lib.optionalString enableStatic ''
    export LDFLAGS="-static -L${talloc}/lib"
  '' + ''
    substituteInPlace GNUmakefile --replace "/usr/local" "$out"
  '';

  sourceRoot = "proot/src";

  meta = with stdenv.lib; {
    homepage = http://proot.me;
    description = "User-space implementation of chroot, mount --bind and binfmt_misc";
    platforms = platforms.linux;
    license = licenses.gpl2;
    maintainers = [ maintainers.ianwookim ];
  };
}
<commit_msg>proot: generate and install manual page<commit_after>{ stdenv, fetchgit, talloc, docutils
, enableStatic ? false }:

stdenv.mkDerivation rec {
  name = "proot-${version}";
  version = "4.0.3";

  src = fetchgit {
    url = "git://github.com/cedric-vincent/proot.git";
    rev = "refs/tags/v${version}";
    sha256 = "95a52b2fa47b2891eb2c6b6b0e14d42f6d48f6fd5181e359b007831f1a046e84";
  };

  buildInputs = [ talloc ];
  nativeBuildInputs = [ docutils ];

  enableParallelBuilding = true;

  preBuild = stdenv.lib.optionalString enableStatic ''
    export LDFLAGS="-static -L${talloc}/lib"
  '';

  makeFlags = [ "-C src" ];

  postBuild = ''
    make -C doc proot/man.1
  '';

  installFlags = [ "PREFIX=$(out)" ];

  postInstall = ''
    install -Dm644 doc/proot/man.1 $out/share/man/man1/proot.1
  '';

  meta = with stdenv.lib; {
    homepage = http://proot.me;
    description = "User-space implementation of chroot, mount --bind and binfmt_misc";
    platforms = platforms.linux;
    license = licenses.gpl2;
    maintainers = [ maintainers.ianwookim ];
  };
}

{'fraction_non_alphanumeric': 0.1485148514851485, 'fraction_numerical': 0.05471599791558103, 'mean_word_length': 3.403669724770642, 'pattern_counts': {'":': 0, '<': 3, '<?xml version=': 0, '>': 3, 'https://': 0, 'lorem ipsum': 0, 'www.': 0, 'xml': 0}, 'pii_count': 0, 'substrings_counts': 0, 'word_list_counts': {'cursed_substrings.json': 0, 'profanity_word_list.json': 0, 'sexual_word_list.json': 0, 'zh_pornsignals.json': 0}}
{'dir': 'git-commits-cleaned', 'id': '6326689', 'n_tokens_mistral': 775, 'n_tokens_neox': 720, 'n_words': 128}